
Part 1: Lists 
Simply put, a list is a variable that contains multiple values separated by a delimiter—the 

default is generally a comma. However, the delimiter can be any single character. Common 

delimiters besides the comma include periods, spaces, semi-colons, tabs (translated to chr(8)), 

carriage returns (chr(10)), and pipes (|). For example, in the list x,y,z a comma is used for the 

delimiter, while 1|2|3|4 uses pipes. Though a list can contain multiple values, a list is still 

considered a "simple" data types. 

 

The position of a value in a list is referred to as an "index" or "position" with the first item in the 

list being at position 1. Individual items within a list are called "elements". 

Creating Lists 

There are several ways of creating a list in ColdFusion. First, of course, is through our normal 

variable setting method, using <cfset>.  It is considered a good practice to start the name of list 

variables with a lowercase "L" or use some other similar convention to clearly denote list 

variables. 

<cfset lNames = "Muffin, Bumpy, Scrappy"> 

<cfset lTypes = "cat;cat;dog"> 

<cfset lColors = "The cat is brown. The other cat is white. The dog is 
black"> 

<cfset FirstNameList = "Larry|Moe|Curly|Shemp"> 

ColdFusion will disregard empty elements in a list. For example, in a list like the one below, 

ColdFusion would only recognize it as having three elements. 

<cfset lNames=",Muffin,Bumpy,,Scrappy,,"> 

If you have lists such as lColors in which you have a delimiter (period) followed by spaces, the 

space will be included in the element's value unless it is also declared a delimiter. You may 

need to use the Trim() function to clean off the space in such lists before using their values. 

 

If multiple form controls use the same name, then their values are stored in a list. 

<form action="form_action.cfm" method="post"> 

<input type="text" name="cust_name" value="cust1"> 

<input type="text" name="cust_name" value="cust2"> 

... 

</form> 

Then on submission, cust_name would have a value of "cust1,cust2". This will also occur 

when using multiple checkboxes with the same name (and multiple items from those 

checkboxes are checked) and when a user selects multiple options in a select box that has the 

"multiple" option enabled. This occurs whether the form uses the post or get method. You can 

also pass lists as values in a URL, ex. http://somesite.tamu.edu?lOptions=1,2,5,6. 

 

ColdFusion will automatically create certain list variables when dealing with forms and queries. 

Specifically:  

 queryname.ColumnList is a list variable containing all the column names returned by a 
query. 

 form.FieldNames is a list variable that contains a list of all the variables in the form 
scope 

List Functions 

ColdFusion offers a variety of built in functions for manipulating lists. While each has different 

arguments which will go over in a moment, keep in mind that empty list elements will always 

be ignored by the system as noted above. If you use any character other than a comma as 

your delimiter, you will have to specify it as an argument in the function call. 

ListAppend(list, value, [delimiter]) 

Add new element to the end of the list 

ListChangeDelims(list, new delimiter, [delimiter]) 

Change the delimiter used in a list 

ListContains(list, string, [delimiter]) 

ListContainsNoCase(list, string, [delimiter]) 

Returns the index of the first element where the requested string was found or 

zero if the string was not found; when using ListContains, the list is basically 

searched as if it were a single string, so it will find any instance of the requested 

string, be it as a list value or something contained within the list! 

ListDeleteAt(list, position, [delimiter]) 

Delete an element from the list at a specified position 

ListFind(list, value, [delimiter]), ListFindNoCase(list, value, [delimiter]) 

Returns the index of the first element in the list matching the specified value or 

zero if not found; unlike ListContains, ListFind requires a perfect value match to 

be considered a match; use ListFindNoCase for case-insensitive comparisons 

ListFirst(list, [delimiter]) 

Returns the first element in the list  

ListGetAt(list, position, [delimiter]) 

Returns the element at the specified position 

ListInsertAt(list, position, [delimiter]) 

Adds a new element to the specified position 

ListLast(list, [delimiter]) 

Returns the last element in the list  

ListLen(list, delimiter]) 

Returns the length of the list, i.e. how many elements it contains 

ListPrepend(list, value, [delimiter]) 

Add new element to the beginning of the list 

ListQualify(list, qualifier, [delimiter], [elements]) 

Wraps elements in a list with a qualifier; can optionally have it do all elements 

(all) or just the alphabetic values (char). 

ListRest(list, [delimiter]) 

Elements in list except for the first element 

ListSetAt(list, position, new value, [delimiter]) 

Change the value of a specified element 

ListSort(list, sort type, [sort order], [delimiter]) 

Sorts a list using specified type and optional order; types available are: numeric 

(sort numbers), text (alphabetical), textnocase (case insensitive alphabetical); 

sort order can be asc (ascending) or desc (descending). 

ListToArray(list, [delimiter], [includeEmptyFields]) 

Convert a list to a single-dimensional array; use optional includeEmptyFields 

parameter to force empty list items to be added as empty array entries; by default 

it does not 

ListValueCount(list, value, [delimiter]) 

ListValueCountNoCase(list, value, [delimiter]) 

Count the number of times a specified value occurs in a list; use 

ListValueCountNoCase() for case-insensitive counting 

ReplaceList(string, list1, list2) 

Replaces occurrences of the elements from a delimited list in a string with 

corresponding elements from another delimited list 

ValueList(queryname.columnname, [delimiter]) 

Converts the values of one column returned in a query to a list 

Looping 

The <cfloop> tag allows you to repeat a block of code until a specific condition is met. With this 

tag, you can perform a variety of loops. The two basic loops most developers are familiar with 

are: 

 An index loop, looking a specific number of times as indicated by the start/finish points, 

with index being the name of a value that is incremented for each loop and step being 

how much the value should be implemented each time (default 1) 

<cfloop  

    index = "parameter name" 

    from = "beginning value" 

    to = "ending value" 

    step = "increment"> 

        HTML or CFML code ... 

</cfloop> 
 

 A conditional loop which continues until a specific condition is met 

<cfloop condition = "expression"> 

    ... 

</cfloop> 

However, ColdFusion also allows you to loop over a range of data types, in which the loops 

runs for each value in the type until completed. You can loop over: 

 Date and time ranges. 
 Lists 
 Queries 
 Arrays 
 Files 
 Structures or collections 

 

We'll cover each of these types of loops as this course progresses. For now, we'll look quickly 

at looping over a date/time range, then looping over a list.  

The basic syntax for looping over a date/time is: 

<cfloop  

    from = "start time" 

    to = "end time" 

    index = "current value" 

    step = "increment"> 

</cfloop> 

Date/Time loops are basically special index loops, where that uses date/times for its from and 

to values, and with step is expressed as a timespan. In this example, we loop over the next 30 

days by 7-day increments. 

<cfset startDate = Now()>  

<cfset endDate = Now() + 30> 

<cfloop from="#startDate#" to="#endDate#" index="i" 
step="#CreateTimeSpan(7,0,0,0)#"> 

 <cfoutput>#dateformat(i, "mm/dd/yyyy")#<br /></cfoutput> 

</cfloop> 

For looping over a list, the syntax is similar: 

<cfloop index="index name" delimiters="item delimiter"  list="list items" 

    ... 

</cfloop> 

The list value can either be a list of items or a list variable (with pound signs). Delimiters are 

optional, with the default being comma. Index in lists is used to store the value of the element 

the specific iteration of the loop is currently on. For example, if we ran this code: 

<cfset ColorList = "red,blue,green,yellow,purple,black"> 

<cfoutput> 

<cfloop list="#ColorList#" index="thiscolor"> 

 This Color: #thiscolor#<br /> 

</cfloop> 

On the first iteration of the loop, thiscolor would equal "red", on the second it would be "blue", 

and so forth. You can also choose to loop over only part of a list by using an index list and the 

ListGetAt() function Using the list above, we could choose to only loop over the first three 

elements like so: 

<cfloop index="thiscolor" from="1" to="2"> 

 This Color: #ListGetAt(ColorList, thiscolor)#<br /> 

</cfloop> 

This would output only the first two colors: 

red 

blue 

Part 2: Queries 
Queries are generally the first complex data type most ColdFusion developers work with, and 

the one they work with the most often. Queries contain rows and columns of data, and can be 

created in several ways: 

 Using <cfquery> to query a database or another query 

 Using specific ColdFusion tags such as <cfsearch>, <cfhttp>, <cfdirectory> which 

generate query data 

 Using query functions such as QueryNew() to create a query object 

Query Functions 

ColdFusion offers eight powerful functions for manipulating queries. In addition to ValueList() 

noted above, the following functions are available: 

isQuery(value) 

Determines if the specified value is a valid query object or not; when used in 

conditional, evaluates to true/false; when outputted, displays Yes or No 

QueryAddColumn(query, column name, [data type], array name) 

Programmatically add a column of values to a query object, populating the cells 

with values from a specified array. While data type is optional, it is always best to 

declare it. If data type isn't declared, ColdFusion must spend extra processing 

time to try to guess the type, which may also have unexpected results or errors. 

QueryAddRow(query, [number of rows]) 

Programmatically adds a blank row to a query object; can optionally declare the 

number of blank rows to add 

QueryConvertForGrid(query, page, number of rows per page) 

Converts a "page" of query data into a structure for use in <cfgrid> controls 

QueryNew(column list, [column data types]) 

Created a new query object; column list and data types are indicated as comma-

delimited lists; available data types are Integer, BigInt, Double, Decimal, Varchar, 

Binary, Bit, Time, and Date 

QuerySetCell(query, column name, new value, row number) 

Used to change the value of a cell at a specified position within a query object 

QuotedValueList(queryname.columnname, [delimiter]) 

Similar to ValueList, except the individual list elements are surrounded by single 

quotes 

While querying a database is the most common way of creating a query object, from the 

functions above you can see that you can create query objects without any database call at all! 

You can also use these functions to manipulate queries from databases to add/change values. 

In this example, we use the QueryAddFunction to run a query that pulls a list of tax exemption 

certificates waiting approval. We then add a new column to this query containing a comma-

delimited list of unpaid orders waiting for the certificate to be approved. 

<!--- run the base query ---> 

<cfquery name="GetCerts" datasource="mydatabase"> 

SELECT exemptionid, clients_external_clientid, datereceived, notes  

FROM clients_external_taxexemptions 

WHERE dateapproved IS NULL 

AND dateexpired IS NULL 

ORDER BY datereceived 

</cfquery> 

 

<!--- create an array to hold the order ids ---> 

<cfset OrderListArray = ArrayNew(1)> 

 

<!--- find orders for each cert and add those to the array as a list---> 

<cfloop query="GetCerts"> 

 <cfquery datasource="#application.dsn#" name="GetAnyPending"> 

  SELECT orderid, dateoforder 

  FROM shopping_orders 

  WHERE clientid = #clients_external_clientid# 

   AND taxexempt = -1 

   AND dateofpayment IS NULL 

  ORDER BY orderid 

 </cfquery> 

  

 <cfset ArrayAppend(OrderListArray, ValueList(GetAnyPending.orderid))> 

</cfloop> 

<!--- add the column ---> 

<cfset QueryAddColumn(GetCerts, "orderlist", "varchar", OrderListArray)> 

Referencing Queries 

You can loop over the contents of a query using <cfoutput> or <cfloop>. In both cases, 

ColdFusion will perform the block of code within the tags for each row in the query. If you are 

displaying data, use <cfoutput>.  

<cfoutput query="queryname"> 

... 

</cfoutput> 

If you are primarily executing logical statements, use <cfloop>. 

<cfloop query="queryname"> 

... 

</cfloop> 

While you cannot nest <cfoutput> tags, except in a special case that we'll discuss in a moment, 

you can nest cfloops within cfoutputs. 

<cfoutput query="q1"> 

#q1.columnname# 

<cfloop query="q2"> 

#q2.columnname# 

</cfloop> 

</cfoutput> 

Sometimes, you might need to programmatically reference very specific data points within a 

query directly. You can reference query cell information through syntactical constructs: 

 queryname.columnname[rownumber] 

 queryname.["columnname"][rownumber] 

Here are two examples of this in use, using an index loop to go over a query and pull values by 

through bracket notation. 

<cfoutput> 

<cfset loopcount = qMyQuery.recordcount> 

<cfloop from="1" to="#variables.loopcount#" index="I"> 

#qMyQuery.columnname[i]# 

</cfloop> 

 

<!--- option two ---> 

<cfset loopcount = qMyQuery.recordcount> 

<cfloop from="1" to="#variables.loopcount#" index="I"> 

#qMyQuery["columnname"][i]# 

</cfloop> 

</cfoutput> 

Grouping Query Output 

When you join two tables in queries that have a master-detail relationship, you will generally 

have column values from your master table repeating in your result set. To suppress this 

repetition in your output of the query, you can utilize the group attribute of the <cfquery> tag. 

<cfoutput query="queryname" group="column to group on"> 

...display repeating values... 

<cfoutput> 

...unique values... 

</cfoutput> 

</cfoutput> 

Notice that only the first output has a query attribute. When using grouping, do not include 

query attributes on the nested queries. Using this, we could turn an output like this: 

 
Figure 1: Query results of a joined query 

To this, without changing our original query or having to do additional queries: 

 
Figure 2: Output of a joined query that is using a group to suppress repeating data 

When using grouping in <cfoutput>, keep the following tips in mind: 

 Include an ORDER BY clause in query to order the data; the first columns ordered 

should be those you want to group, otherwise your grouping will not work correctly 

 If you need to do multiple levels of grouping, use nested <cfoutput> statements, with 

each statement having a group attribute; these groupings should follow the ordering of 

the columns in the query's ORDER BY clause 

Query of Queries 

Sometimes, you may need to reuse the data or a subset of data from a query, such as 

presenting it in another format, changing the sorting, or performing some other basic SQL 

functions on it. Rather than requiring the database with a minor change, you can use 

ColdFusion to query and existing query using a subset of standard SQL options. Some 

situations in which you might want to query a query include: 

 Reordering data - when you need to present or summarize an existing recordset in 

different ways 

 Drill-down and Next-n interfaces - by caching a query of all of the records, you can then 

use the existing query to perform the drills downs while avoiding additional calls to the 

database, which improves performance 

 Combining records sets - you can perform joins and unions on records from different 

data sources to create one single query 

 Graphing data - you can re-query the data to find and graph subsets and summaries of 

the data 

 Processing uploaded data - you can uploaded a structured file into a query variable, 

then use a query of a query to further process and manipulate the information 

To perform a query of a query, you use the dbtype attribute of the <cfquery> tag instead of the 

datasource attribute. In the SQL's FROM statement, use the name of the existing query(s) for 

the table name(s). 

<cfquery name="qArticlesByTitle" datasource="CF8advOnYourOwn"> 

SELECT articleid, title, updatedate, author 

FROM articles 

ORDER BY title 

</cfquery> 

<cfquery name="qArticlesByUpdateDate" dbtype="query"> 

SELECT articleid, title, updatedate, author 

FROM qArticlesByTitle 

ORDER BY updatedate desc 

</cfquery> 

When performing a query of a query, you can use a subset of SQL statements, functions, and 

operators, which covers the most used items found in queries. In addition to being able to use 

dot notation and aliases, you can use the following SQL options: 

 Statements: SELECT, WHERE, GROUP BY, HAVING, ORDER BY, AS, UNION 

 Operators: AND, OR, LIKE, IN, BETWEEN, IS NULL, NOT, <=, >=, =<>, <>, != 

 Functions: COUNT(), SUM(), AVG(), MAX(), MIN(), CAST() 

There is one limit to performing queries of queries, however. Joining of tables is limited to 

simple joins in the WHERE statement, rather than proper INNER JOIN/OUTER JOIN 

statements.  

Part 3: Transactions 
A transaction is a logical unit of work against a database that either succeeds or fails as a 

whole. If all parts of it succeed, then all of the work is completed. If any one part fails, the entire 

unit is rolled back as if it never happened. Transactions are used when performing multiple 

actions against a database that are related or dependent on one another. During the 

transaction, the referenced tables are set to read only mode for all others attempting to access 

it, preventing others from changing the data. One common example of when a transaction is 

used would be when recording a transfer of funds from a savings account to a checking 

account. This requires two transactions: 

 Withdrawal from savings 

 Deposit to checking 

Without a transaction, if the deposit fails, but the withdrawal is successful, the customer loses 

money. Using transactions, if the deposit fails, the whole process is rolled back so withdrawal 

isn't taken either. To group queries into a transaction, use the <cftransaction> tag. 

<cftransaction isolation="read_uncommitted|read_committed|repeatable_read"> 

[block of queries] 

</cftransaction> 

The isolation attribute is an optional way of determining what kind of read transactions can 

occur on the tables being queried during the transaction. ColdFusion recognizes four isolation 

levels: 

 read_uncommitted: Allows dirty read1, non-repeatable read2, and phantom3 

 read_committed: Allows non-repeatable read and phantom. Does not allow dirty read. 

 repeatable_read: Allows phantom. Does not allow dirty read or non-repeatable read. 

 serializable: Does not allow dirty read, non-repeatable read, or phantom. 

If you do not specify an isolation attribute, ColdFusion will use the default setting from your 

database. Keep in mind that not all databases support all four of these options, so you'll need 

to look at your database's documentation to find out which types of isolation are supported and 

what your database's default is. 

<cftransaction> 

<cfquery datasource="Banking"> 

UPDATE Savings 

SET Balance=Balance-200 

WHERE Account_No=193847 

</cfquery> 

<cfquery datasource="Banking"> 

UPDATE Checking 

SET Balance=Balance+200 

WHERE Account_No=C193847 

</cfquery> 

</cftransaction> 

Queries within the transaction cannot span multiple data sources. A transaction can have four 

possible results: 

1. Commit - all of the queries were successful and all changes become permanent 

2. Rollback - an error occurs within the <cftransaction> tag, and all database changes are 

reversed 

3. Partial commit and rollback - starting in ColdFusion 7, an ACTION attribute was added 

that allows you to specify points in the transaction during which a partial commit or 

rollback can be performed; this allows you batch process multiple related groups of 

 
1 A dirty read allow a second SQL transaction to read a row before the first SQL transaction executes a COMMIT 
2 A non-repeatable read occurs when an SQL transaction reads a row and then a second SQL transaction 
modifies or deletes the row and executes a COMMIT 
3 In a phantom read, an SQL transaction reads rows that meet search criteria, a second SQL transaction then 
modifies at least one row that meets the first transaction's search criteria; the first transaction repeats the search, 
resulting in a different result set 

queries within one transaction block, and is primarily only used when the queries are 

highly interdependent. 

4. Starting in ColdFusion 8, a new attribute is available called SAVEPOINT which is used 

in conjunction with the ACTION attribute to allow you to designate a point within a 

transaction that you want to rollback to. If you had an insert, update, and delete 

statement, you could add save points to each, then in tour error handling code you 

could determine if you need to rollback each statement on an individual basis 

In this example, we run several sets of deposits and withdrawals, committing between sets. 

<cftransaction> 

<cfquery datasource="Banking"> 

UPDATE Savings 

SET Balance=Balance-200 

WHERE Account_No=193847 

</cfquery> 

<cfquery datasource="Banking"> 

UPDATE Checking 

SET Balance=Balance+200 

WHERE Account_No=C193847 

</cfquery> 

<cftransaction action="commit" /> 

<cfquery datasource="Banking"> 

UPDATE Savings 

SET Balance=Balance-200 

WHERE Account_No=1938470 

</cfquery> 

<cfquery datasource="Banking"> 

UPDATE Checking 

SET Balance=Balance+200 

WHERE Account_No=C1938470 

</cfquery> 

</cftransaction> 

We can use the action and savepoint attributes to process interdependent statements within a 

transaction, giving even more finite control over what is rolled back and why. For example, if 

the withdrawal from the Savings account would bring the account balance below zero, then 

we'd want to selectively rollback the withdrawal while still recording the attempt (and perhaps 

adding a transaction to charge an overdraft fee). 

<cftransaction> 

<cftransaction action="setsavepoint" savepoint="trans1"> 

<cfquery datasource="Banking"> 

UPDATE Savings 

SET Balance=Balance-200 

WHERE Account_No=193847 

</cfquery> 

<cfquery name="SaveBal" datasource="Banking"> 

SELECT Balance 

FROM Savings 

WHERE Account_No = 193847 

</cfquery> 

<cfif SaveBal.Balance LT 0> 

<cftransaction action="rollback" savepoint="trans1"> 

<cfelse> 

<cfquery datasource="Banking"> 

UPDATE Checking 

SET Balance=Balance+200 

WHERE Account_No=C193847 

</cfquery> 

</cfif> 

...other processing to record transaction... 

</cftransaction> 

Part 4: Data Importing With CFFILE 
When developing applications, you may be tasked with importing and processing structured 

data from external or legacy systems. Usually this structured data will be in a delimited file, in 

which some delimiter (usually a comma or a tab) separates each field and each record is 

separated by a carriage return (which ColdFusion represent by invoking the CHR(10) function).  

 
Figure 3: Example of a structured, comma-delimited file 

Files can be uploaded to the server using ColdFusion HTML or XML forms, which allow a user 

to select a single file from their local machine/network and sent it, along with the rest of the 

form information, to the ColdFusion for processing with the <cffile> tag. You can use <cffile> to 

upload almost any file type, including binary files, images, Word files, PDFs, zip files, etc. File 

uploading can cause a significant, though temporary, drain on system resources as server 

memory is allocated in equal proportions to the size of the file being uploaded. For example, if 

a user attempts to upload a 100 MB file, it will consume 100 MB of RAM for the duration of the 

request. 

 

You can use memory-throttling settings in the ColdFusion 8 Administrator to restrict the 

amount of data, which can be posted during a single request, and to throttle requests to avoid 

running out of system resources. These settings should be utilized, in particular, to help 

prevent denial-of-service attacks in which a rogue program attempts to upload a number of 

large files concurrently, thereby consuming all of the system memory. You can also disable file 

uploading all together in the Administrator, however this generally is unnecessary and is more 

of a hindrance to development than useful security option. You can also use the 

CGI.content_length variable to get a rough idea of the size of a pending file and skip 

processing it all together if it is over a certain size. This variable contains the size of the file, in 

bytes, plus the length of any other request content. 

Uploading Files 

When you are adding a file upload field to a form, you much set the form's ENCTYPE attribute 

to "multipart/form-data" so the browser will recognize that it will be passing a file to the server. 

The file field itself it set using a <cfinput> with a type of "file". 

<cfform enctype="multipart/form-data"> 

Select file to upload: <cfinput type="file" name="datafile"><br /><br 
/> 

<cfinput type="submit" name="btnSubmit" value="Upload File"> 

</cfform> 

 
Figure 4: Example of a file upload form 

In your processing code, you would use the <cffile> tag to take the file and write it to a specific 

location. For uploading, the basic syntax is: 

<cffile  

    action = "upload" 

    destination = "full pathname" 

    fileField = "form field" 

    accept = "MIME type|file type" 

    nameConflict = "behavior" 

    > 

Destination should be either an absolute path, or a path relative to the ColdFusion temporary 

directory. You can use the accept attribute to give a comma-delimited lists of acceptable file 

types, to limit what types of files may be uploaded. For example, if you have a field allowing 

users to upload an avatar, you'd want to limit the field to image files. For example, in this 

instance of <cffile> we are only allowing JPEG and plain text files to be uploaded. 

<cffile action="upload" 

filefield="datafile" 

destination="/my/directory/" 

nameconflict="makeunique" 

accept="image/jpeg,text/plain"> 

The NameConflict attribute allows you to instruct ColdFusion on how it should behave if a file 

with the same name exists in the upload destination. It has four possible options: 

1. Error: file is not saved and ColdFusion returns an error message 

2. Skip: the file is not saved but the rest of the page continues processing; this allow you to 

perform custom behavior based on file properties 

3. Overwrite: replaces the existing file 

4. MakeUnique: gives the file a unique name by adding a number to the file name 

CFFILE variables 

Once the file is uploaded, the CFFILE scope will be created containing a variety of variables 

containing information about the file.  
Parameter Description 
AttemptedServerFile Initial name ColdFusion used when it was attempting to save the file 

ClientDirectory Directory location of the file uploaded from the client's system 

ClientFile Name of the file uploaded from the client's system 

ClientFileExt Extension of the uploaded file on the client system (without a period) 

ClientFileName Name of the uploaded file on the client system (without an extension) 

ContentSubType MIME content subtype of the saved file 

ContentType MIME content type of the saved file 

DateLastAccessed Date and time the uploaded file was last accessed 

FileExisted Whether the file already existed with the same path (yes or no) 

FileSize Size of the uploaded file 

FileWasAppended Whether ColdFusion appended uploaded file to a file (yes or no) 

FileWasOverwritten Whether ColdFusion overwrote a file (yes or no) 

FileWasRenamed Whether uploaded file renamed to avoid a name conflict (yes or no) 

FileWasSaved Whether ColdFusion saves a file (yes or no) 

OldFileSize Size of a file that was overwritten in the file upload operation 

ServerDirectory Directory of the file saved on the server 

ServerFile Filename of the file saved on the server 

ServerFileExt Extension of the uploaded file on the server (without a period) 

ServerFileName Name of the uploaded file on the server (without an extension); if you use 

the makeunique option, then use this variable to determine the file's final 

name 

TimeCreated Time the uploaded file was created 

TimeLastModified Date and time of the last modification to the uploaded file 

 

If you are performing multiple uploads on a page, CFFILE will only contain the information for 

the last upload performed. However, you can use the "result" attributes of the <cffile> upload 

tags to give each one a unique name, which you can then use to reference the individual file's 

CFFILE variables using resultname.variablename. 

Parsing a Data File 

Prior to ColdFusion 8, to parse a file on the server, you'd have to use another instance of 

<cffile> to read the file, similar to this: 

<cfoutput> 

<cffile action="read" 

file="c:\coldfusion8\...\contacts.txt" variable="thisContent"> 

<cfloop list="#thiscontent#" index="thisRow" delimiters="#chr(10)#"> 

<cfif listlen(thisRow) is 4> 

<cfloop list="#thisRow#" delimiters="," index="thiscolumn"> 

#thiscolumn# <br /> 

</cfloop> 

</cfif> 

</cfloop> 

</cfoutput> 

However, you can now parse structured text files using the <cfloop> tag to loop over the file. 

The index attribute contains one row from the file. When using <cfloop> with files, you must 

use an absolute path to point to the file. Here is the same code above, but now using cfloop to 

parse the file instead. 

<cfoutput> 

<cfloop file="c:\coldfusion8\...\contacts.txt" index="thisRow"> 

<cfif listlen(thisRow) is 4> 

<cfloop list="#thisRow#" delimiters="," index="thiscolumn"> 

#thiscolumn# <br /> 

</cfloop> 

</cfif> 

</cfloop> 

</cfoutput> 

When dealing with uploaded data, you might also want to sort it, filter it, etc, before you insert it 

into a database, or use it in some other fashion. One very useful way of doing this is by 

converting the file contents to a query object, using some of the query functions we discussed 

before. 

<!--- process file into a <cfquery> object ---> 

<cfset lcolumns = "firstname,lastname,zipcode,cellphone"> 

 

<!--- Create the new query object ---> 

<cfset qContacts=querynew(variables.lColumns, 

"varchar,varchar,integer,varchar")> 

 

<!--- Open the uploaded file directly using the file attribute of the 
<cfloop> tag ---> 

<cfloop file="#cffile.serverdirectory#\#cffile.serverfile#" index="thisrow"> 

<cfset column = 0> 

<cfset queryaddRow(qContacts,1)>  

<cfset nRow = qContacts.recordcount> 

 

<!--- Loop through a single row of the file and create a single row of 
qContacts ---> 

<cfloop list="#variables.thisrow#" delimiters="," index="thiscell"> 

<cfset column = column + 1> 

<cfset columnName=listgetat(lColumns,column)> 

<cfset querysetcell(qContacts, variables.columnName, 
thisCell,nRow)> 

</cfloop> 

</cfloop> 

<!--- end text file processing ---> 

We can then perform queries against the new object, like this: 

<cfquery name="qSort" dbtype="query"> 

SELECT lastname,firstname 

FROM qContacts 

ORDER by lastname 

</cfquery> 

Best Practices and Reminders 
1. When using list variables, give them distinctive names that clearly identify them as lists, 

such as having the first letter of the name set to a lowercase L, or some other similar 
method. 

2. ListContains()/ListContainsNoCase() looks for values within each element, while 
ListFind()/ListFindNoCase() searches for elements that exactly match the requested 
values. 

3. List positions start at position 1. 

4. ColdFusion allows you to loop over a wide range of data types using special variants of 
index loops. When looping, "index" is the variable used to store the value of the current 
loop iteration.  

5. Query objects can be created in a variety of ways, not just by using <cfquery> 

6. When using QueryAddColumn, always declare data type of the column to ensure 
accuracy and improve performance. 

7. Use <cfoutput> to loop over a query when you want to actually display content; use 
<cfloop> when you are primarily performing logical statements. 

8. <cfoutput> tags cannot be nested, except when using grouping. When grouping 
<cfoutput>, sort the fields in the SQL query in the same order they will be grouped. 

9. Use query of queries to reuse data while improving performance and reducing the 
number of database calls. 

10. Use transactions when performing multiple, related actions against a database to 
rollback queries if part of the set fails; use partial commits and savepoints for more finite 
control over what is rolled back. 

11. Use HTML/XML ColdFusion forms with <cffile> to allow file uploading in forms. During 
the upload, an equivalent amount of RAM will be consumed relative to the size of the 
file. 

12. Use ColdFusion's throttling and file limitation settings to prevent denial of service 
attacks and the consumption of all of your server's memory during normal operations. 

13. Forms with file uploading must use an enctype of "multipart/form-data" 

14. Use the result attribute in <cffile> when performing multiple file uploads on a single 
page to enable access to each ones individual CFFILE structure 

15. To parse a structured data file, use <cfloop> to loop over the file contents one row at a 
time. Convert the file to a query structure if you need to manipulate the data before 
inserting it into a database. 

Applying What We Learned 
1. Given this list:  

<cfset CountrySingerList = "Rhett Akins,Alabama,Gary Allan,John Anderson,Jessica 
Andrews,Sherrie Austin,David Ball,The Bellamy Brothers,John Berry,Clint 
Black,Blackhawk,Chad Brock,Garth Brooks,Brooks and Dunn,Tracy Byrd,Chris 
Cagle,Jeff Carson"> 

a. Try using some of the various list functions, in particular: 
i. Add an item to the list 
ii. Remove an item from the middle of the list 
iii. Sort the list 
iv. Change your delimiter 
v. Convert a list to an array 
vi. Convert a query result to a list 

b. Loop over the entire list 
c. Loop over the middle 3 items in the list 

2. Queries 
a. Use query functions 

i. Create a query object 
ii. Add a row 
iii. Add a column 
iv. Set a cell value 

b. Reference a query using <cfloop> and <cfoutput> as is appropriate 
c. Create a query using joins and then use grouping to cleanly output the results 
d. Perform a query of a query 

3. Try using transactions and deliberately cause them to fail. Watch the results in the 
database to see how rollback works. 

4. Given a structured data file 
a. Read the file 
b. Parse its contents and display them in clean format 
c. Put into a query structure and summarize the contents 

